

Supplemental Material for  
Cross-Lagged Panel Modeling  
with Binary and Ordinal Outcomes

Bengt Muthén, Tihomir Asparouhov  
& Katie Witkiewitz \*

April 6, 2024

---

\*We thank Noah Hastings who provided support with tables and figures.

# 1 Mplus input scripts

Mplus input for key analyses.

Table 1: Univariate analysis of abstinence using unrestricted model

```
USEVARIABLES = z1-z8;  
CATEGORICAL = z1-z8;  
  
DEFINE:      CUT z1-z8 (0.5);  
  
ANALYSIS:    ESTIMATOR = BAYES;  
              BITERATIONS = (5000);  
              THIN = 10;  
              PROCESSORS = 8;  
  
MODEL:       z1-z8 WITH z1-z8;  
  
OUTPUT:      STANDARDIZED RESIDUAL TECH8 TECH10;  
  
PLOT:        TYPE = PLOT3;
```

Table 2: Univariate analysis of abstinence using RI-AR2

```
USEVARIABLES = z1-z8;  
CATEGORICAL = z1-z8;  
  
DEFINE:      CUT z1-z8 (0.5);  
  
ANALYSIS:    ESTIMATOR = BAYES;  
              BITERATIONS = (5000);  
              THIN = 10;  
              PROCESSORS = 8;  
  
MODEL:       i BY z1-z8@1;  
              z2^-z8^ PON z1^-z7^;  
              z3^-z8^ PON z1^-z6^;  
  
OUTPUT:      STANDARDIZED RESIDUAL TECH8 TECH10;  
  
PLOT:        TYPE = PLOT3;
```

Table 3: Bivariate analysis of stress and abstinence using RI-CLPM2

```

USEVARIABLES = y2-y9 z2-z9; ! y is stress, z is alcohol risk
CATEGORICAL = z2-z9;

DEFINE:      CUT z2-z9(0.5);

ANALYSIS:   ESTIMATOR = BAYES;
             BITERATIONS = (5000);
             PROCESSORS = 8;
             THIN = 10;

MODEL:      iz BY z2-z9@1;
             z3^-z9^ PON on z2^-z8^;
             z4^-z9^ PON z2^-z7^;
             iy BY y2-y9@1;
             y3^-y9^ PON y2^-y8^;
             y4^-y9^ PON y2^-y7^;

             y3^-y9^ PON z2^-z8^;
             z3^-z9^ PON y2^-y8^;

             y2^-y9^ PWITH z2^-z9^;

OUTPUT:     STANDARDIZED RESIDUAL TECH8 TECH10;

PLOT:       TYPE = PLOT3;

```

Table 4: Bivariate analysis of stress and abstinence using RI-RCLPM

```

USEVARIABLES = y2-y9 z2-z9; ! y is stress, z is alcohol risk
CATEGORICAL = z2-z9;

DEFINE:          CUT z2-z9(0.5);

ANALYSIS:        ESTIMATOR = WLSMV;
                  PARAMETERIZATION = THETA;
                  STARTS = 20;
                  BOOTSTRAP = 500;

MODEL:           iz BY z2-z9@1;
                  z3^-z9^ PON z2^-z8^;
                  z4^-z9^ PON z2^-z7^;
                  iy by y2-y9@1;
                  y3^-y9^ PON y2^-y8^;
                  y4^-y9^ PON y2^-y7^;

                  y3^-y9^ PON z2^-z8^;
                  z3^-z9^ PON y2^-y8^;

                  y3^-y9^ PON z3^-z9^ (ryz);
                  z3^-z9^ PON y3^-y9^ (rzy);

                  y2^ WITH z2^;

MODEL CONSTRAINT: ! Restriction (a):
                  0<ryz*rzy;
                  0<1-ryz*rzy;

OUTPUT:          STANDARDIZED RESIDUAL TECH1 TECH10;

                  CINTERVAL(BOOTSTRAP);

PLOT:           TYPE = PLOT3;

```

Table 5: Bivariate analysis of stress and abstinence using RI-RLPM

```

USEVARIABLES = y2-y9 z2-z9; ! y is stress, z is alcohol risk
CATEGORICAL = z2-z9;

DEFINE:      CUT z2-z9(0.5);

ANALYSIS:    ESTIMATOR = WLSMV;
              PARAMETERIZATION = THETA;
              STARTS = 20;
              BOOTSTRAP = 500;

MODEL:       iz BY z2-z9@1;
              z3^-z9^ PON z2^-z8^;
              z4^-z9^ PON z2^-z7^;
              iy BY y2-y9@1;
              y3^-y9^ PON y2^-y8^;
              y4^-y9^ PON y2^-y7^;

              ! y3^-y9^ PON z2^-z8^;
              ! z3^-z9^ PON y2^-y8^;

              y3^-y9^ PON z3^-z9^ (ryz);
              z3^-z9^ PON y3^-y9^ (rzy);

              y2^ WITH z2^;

              y3^-y9^ PWITH z3^-z9^;

OUTPUT:      STANDARDIZED RESIDUAL TECH1 TECH10;
              CINTERVAL(BOOTSTRAP);

PLOT:        TYPE = PLOT3;

```

Table 6: Bivariate analysis of stress and abstinence using single-direction lag 0

```

USEVARIABLES = y2-y9 z2-z9; ! y is stress, z is alcohol risk
CATEGORICAL = z2-z9;

DEFINE:      CUT z2-z9(0.5);

ANALYSIS:    ESTIMATOR = WLSMV;
              PARAMETERIZATION = THETA;
              STARTS = 20;

MODEL:       iz BY z2-z9@1;
              z3^-z9^ PON z2^-z8^;
              z4^-z9^ PON z2^-z7^;
              iy by y2-y9@1;
              y3^-y9^ PON y2^-y8^;
              y4^-y9^ PON y2^-y7^;

              y3^-y9^ PON z2^-z8^;
              z3^-z9^ PON y2^-y8^;

              ! y2^-y9^ PWITH z2^-z9^;

              y2^ WITH z2^;
              y3^-y9^ PON z3^-z9^ (lag0);

OUTPUT:      STANDARDIZED RESIDUAL TECH1 TECH10;

PLOT:        TYPE = PLOT3;

```

Table 7: Univariate analysis of alcohol risk using unrestricted regular ordinal probit

```

USEVARIABLES = z1-z8;
CATEGORICAL = z1-z8;

ANALYSIS:    ESTIMATOR = BAYES;
              BITERATIONS = (5000);
              THIN = 10;
              PROCESSORS = 8;

MODEL:       z1-z8 WITH z1-z8;

OUTPUT:      STANDARDIZED RESIDUAL TECH8 TECH10;

PLOT:        TYPE = PLOT3;

```

Table 8: Univariate analysis of alcohol risk using regular ordinal probit RI-AR2

```

USEVARIABLES = z1-z8;
CATEGORICAL = z1-z8;

ANALYSIS:    ESTIMATOR = BAYES;
              BITERATIONS = (5000);
              THIN = 10;
              PROCESSORS = 8;

MODEL:       i BY z1-z8@1;
              z2^-z8^ PON z1^-z7^;
              z3^-z8^ PON z1^-z6^;

OUTPUT:      STANDARDIZED RESIDUAL TECH8 TECH10;
PLOT:        TYPE = PLOT3;

```

Table 9: Univariate analysis of alcohol risk using unrestricted two-part ordinal

```

USEVARIABLES = u1-u8 p1-p8;
CATEGORICAL = u1-u8 p1-p8;

DATA TWOPART: NAMES = z1-z8;
               BINARY = u1-u8;
               CONTINUOUS = p1-p8;
               CUTPOINT = 0;
               TRANSFORM = NONE;

ANALYSIS:     ESTIMATOR = BAYES;
               BITERATIONS = (10000);
               THIN = 10;
               PROCESSORS = 8;

MODEL:        u1-u8 WITH u1-u8;
               p1-p8 WITH p1-p8;
               p1^-p8^ ON u1^-u8^;
               p1^-p8^ PON u1^-u8^@0;

OUTPUT:       STANDARDIZED RESIDUAL TECH8 TECH10;

PLOT:         TYPE = PLOT3;

```

Table 10: Univariate analysis of alcohol risk using two-part ordinal RI-AR2

```

USEVARIABLES = u1-u8 p1-p8;
CATEGORICAL = u1-u8 p1-p8;

DATA TWOPART:  NAMES = z1-z8;
                BINARY = u1-u8;
                CONTINUOUS = p1-p8;
                CUTPOINT = 0;
                TRANSFORM = NONE;

ANALYSIS:      ESTIMATOR = BAYES;
                ITERATIONS = (10000);
                THIN = 10;
                PROCESSORS = 8;

MODEL:         ib BY u1-u8@1;
                ip BY p1-p8@1;
                u2^-u8^ PON u1^-u7^;
                u3^-u8^ PON u1^-u6^;
                p2^-p8^ PON p1^-p7^;
                p3^-p8^ PON p1^-p6^;
                ! u1-u5 PWITH p1-p5;

OUTPUT:        STANDARDIZED RESIDUAL TECH8 TECH10;

PLOT:          TYPE = PLOT3;

```



Table 11: Bivariate analysis of stress and alcohol risk using two-part ordinal RI-CLPM

```

USEVARIABLES = y1-y8 u1-u8 p1-p8;
CATEGORICAL = u1-u8 p1-p8;

DATA TWOPART:  NAMES = z1-z8;
                BINARY = u1-u8;
                CONTINUOUS = p1-p8;
                CUTPOINT = 0;
                TRANSFORM = none;

ANALYSIS:      ESTIMATOR = BAYES;
                BITERATIONS = (10000);
                ! FBITERATIONS = 200;
                THIN = 10;
                PROCESSORS = 8;
                ALGORITHM = GIBBS(RW);

MODEL:         iy BY y1-y8@1;
                ib BY u1-u8@1;
                ip BY p1-p8@1;

                ! univariates:
                y2^-y8^ PON y1^-y7^;
                y3^-y8^ PON y1^-y6^;

                u2^-u8^ PON u1^-u7^;
                u3^-u8^ PON u1^-u6^;

                p2^-p8^ PON p1^-p7^;
                p3^-p8^ PON p1^-p6^;

                ! bivariate:
                y2^-y8^ PON u1^-u7^;
                y2^-y8^ PON p1^-p7^;

                u2^-u8^ PON y1^-y7^;
                p2^-p8^ PON y1^-y7^;

                ! covariances:
                y1^-y8^ PWITH u1^-u8^;
                y1^-y8^ PWITH p1^-p8^;

OUTPUT:        STANDARDIZED RESIDUAL TECH8 TECH10;
PLOT:          TYPE = PLOT3;

```

Table 12: Bivariate analysis of stress and alcohol risk treatment effects using two-part ordinal RI-CLPM

```

USEVARIABLES = y1-y8 u1-u8 p1-p8;
CATEGORICAL = u1-u8 p1-p8;
CLASSES = c(9);
KNOWNCLASS = c(cCell = 1-9);

DATA TWOPART:
  NAMES = z1-z8;
  BINARY = u1-u8;
  CONTINUOUS = p1-p8;
  CUTPOINT = 0;
  TRANSFORM = NONE;

ANALYSIS:
  TYPE = MIXTURE;
  ESTIMATOR = BAYES;
  BITERATIONS = (5000);
  THIN = 10;
  PROCESSORS = 8;
  ALGORITHM = GIBBS(RW);

MODEL:
  %OVERALL%
  iy BY y1-y8@1;
  ib BY u1-u8@1;
  ip BY p1-p8@1;

  ! univariates:
  y2^-y8^ PON y1^-y7^;
  y3^-y8^ PON y1^-y6^;
  u2^-u8^ PON u1^-u7^;
  u3^-u8^ PON u1^-u6^;
  p2^-p8^ PON p1^-p7^;
  p3^-p8^ PON p1^-p6^;

  ! bivariate:
  y2^-y8^ PON u1^-u7^;
  y2^-y8^ PON p1^-p7^;
  u2^-u8^ PON y1^-y7^;
  p2^-p8^ PON y1^-y7^;

  ! covariances:
  y1^-y8^ PWITH u1^-u8^;
  y1^-y8^ PWITH p1^-p8^;

  %c#1%
  [iy@0 ib@0 ip@0];
  %c#9%
  [iy ib ip];

OUTPUT:
  STANDARDIZED RESIDUAL TECH8 TECH10;

PLOT:
  TYPE = PLOT3;

```